Rules for integrands of the form $\left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n$

1. $\int\left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d$

   1. $\int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e == c^2 d$

      x: $\int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e == c^2 d$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

- Basis: $\frac{F[\operatorname{ArcSinh}[c x]]}{\sqrt{1+c^2 x^2}} == \frac{1}{c} \operatorname{Subst}[F[x], x, \operatorname{ArcSinh}[c x]] \, \partial_x \operatorname{ArcSinh}[c x]$

- Note: When $n == 1$, this rule would result in a slightly less compact antiderivative since $\int (a+b x)^n \, dx$ returns a sum.

- Rule: If $e == c^2 d$, then

$$\int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx \rightarrow \frac{\sqrt{1+c^2 x^2}}{c \sqrt{d+e x^2}} \operatorname{Subst}\left[\int (a+b x)^n \, dx, x, \operatorname{ArcSinh}[c x]\right]$$

- Program code:

```
(* Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/c*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*Subst[Int[(a+b*x)^n,x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] *)
```

1: $\int \frac{1}{\sqrt{d+e x^2}\,(a+b\,\text{ArcSinh}[c\,x])}\,dx$ when $e == c^2 d$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

- Rule: If $e == c^2 d$, then

$$\int \frac{1}{\sqrt{d+e x^2}\,(a+b\,\text{ArcSinh}[c\,x])}\,dx \longrightarrow \frac{\sqrt{1+c^2 x^2}}{b\,c\,\sqrt{d+e x^2}}\,\text{Log}[a+b\,\text{ArcSinh}[c\,x]]$$

- Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])),x_Symbol] :=
  1/(b*c)*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*Log[a+b*ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d]
```

2: $\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \land n \neq -1$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e\,x^2}} == 0$

Rule: If $e == c^2 d \land n \neq -1$, then

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\sqrt{1+c^2 x^2}}{b\,c\,(n+1)\,\sqrt{d+e\,x^2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(b*c*(n+1))*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*(a+b*ArcSinh[c*x])^(n+1) /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && NeQ[n,-1]
```

2. $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n > 0$

1: $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2 d \land p \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $e == c^2 d \land p \in \mathbb{Z}^+$, let $u = \int (d+e\,x^2)^p\,dx$, then

$$\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSinh}[c\,x]) - b\,c\int \frac{u}{\sqrt{1+c^2 x^2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

2. $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n > 0 \land p > 0$

1: $\int \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n > 0$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $e == c^2 d \land n > 0$, then

$$\int \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{x\sqrt{d+e x^2}\,\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{2}-\frac{b\,c\,n\sqrt{d+e x^2}}{2\sqrt{1+c^2 x^2}}\int x\,\left(a+b\,\text{ArcSinh}[c\,x]\right)^{n-1}dx+\frac{\sqrt{d+e x^2}}{2\sqrt{1+c^2 x^2}}\int\frac{\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{\sqrt{1+c^2 x^2}}dx$$

Program code:

```
Int[Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/2 -
  b*c*n/2*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[x*(a+b*ArcSinh[c*x])^(n-1),x] +
  1/2*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0]
```

2: $\int\left(d+e\,x^2\right)^p\left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx$ when $e==c^2 d\ \land\ n>0\ \land\ p>0$

Derivation: Inverted integration by parts and piecewise constant extraction

Basis: If $e==c^2 d$, then $\partial_x\frac{\left(d+e\,x^2\right)^p}{\left(1+c^2 x^2\right)^p}==0$

Rule: If $e==c^2 d\ \land\ n>0\ \land\ p>0$, then

$$\int\left(d+e\,x^2\right)^p\left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx\longrightarrow$$

$$\frac{x\left(d+e\,x^2\right)^p\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{2\,p+1}+$$

$$\frac{2\,d\,p}{2\,p+1}\int\left(d+e\,x^2\right)^{p-1}\left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx-\frac{b\,c\,n\left(d+e\,x^2\right)^p}{(2\,p+1)\left(1+c^2 x^2\right)^p}\int x\left(1+c^2 x^2\right)^{p-\frac{1}{2}}\left(a+b\,\text{ArcSinh}[c\,x]\right)^{n-1}dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n/(2*p+1)*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0]
```

3. $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p < -1$

1: $\int \frac{(a+b \text{ArcSinh}[c x])^n}{(d+e x^2)^{3/2}} \, dx$ when $e == c^2 d \land n > 0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{1}{(d+e x^2)^{3/2}} == \partial_x \frac{x}{d \sqrt{d+e x^2}}$

Basis: $\partial_x (a+b \text{ArcSinh}[c x])^n == \frac{b c n (a+b \text{ArcSinh}[c x])^{n-1}}{\sqrt{1+c^2 x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

Rule: If $e == c^2 d \land n > 0$, then

$$\int \frac{(a+b \text{ArcSinh}[c x])^n}{(d+e x^2)^{3/2}} \, dx \to \frac{x (a+b \text{ArcSinh}[c x])^n}{d \sqrt{d+e x^2}} - \frac{b c n \sqrt{1+c^2 x^2}}{d \sqrt{d+e x^2}} \int \frac{x (a+b \text{ArcSinh}[c x])^{n-1}}{1+c^2 x^2} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*ArcSinh[c*x])^n/(d*Sqrt[d+e*x^2]) -
  b*c*n/d*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*Int[x*(a+b*ArcSinh[c*x])^(n-1)/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0]
```

2: $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p < -1 \land p \neq -\frac{3}{2}$

Rule: If $e == c^2 d \land n > 0 \land p < -1 \land p \neq -\frac{3}{2}$, then

$$\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx \to$$

$$-\frac{x\left(d+e x^2\right)^{p+1}\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{2\,d\,(p+1)}+$$
$$\frac{2\,p+3}{2\,d\,(p+1)}\int\left(d+e\,x^2\right)^{p+1}\left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx+\frac{b\,c\,n\left(d+e\,x^2\right)^p}{2\,(p+1)\left(1+c^2 x^2\right)^p}\int x\left(1+c^2 x^2\right)^{p+\frac{1}{2}}\left(a+b\,\text{ArcSinh}[c\,x]\right)^{n-1} dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] +
  b*c*n/(2*(p+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[x*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

4: $\int\frac{\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{d+e\,x^2}\,dx$ when $e==c^2 d\ \land\ n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $e==c^2 d$, then $\frac{1}{d+e\,x^2}==\frac{1}{c\,d}\,\text{Subst}[\text{Sech}[x],\,x,\,\text{ArcSinh}[c\,x]]\,\partial_x\text{ArcSinh}[c\,x]$

Note: If $n\in\mathbb{Z}^+$, then $(a+b\,x)^n\,\text{Sech}[x]$ is integrable in closed-form.

Rule: If $e==c^2 d\ \land\ n\in\mathbb{Z}^+$, then

$$\int\frac{\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{d+e\,x^2}\,dx\ \longrightarrow\ \frac{1}{c\,d}\,\text{Subst}\left[\int(a+b\,x)^n\,\text{Sech}[x]\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/(c*d)*Subst[Int[(a+b*x)^n*Sech[x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0]
```

3. $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1$

1: $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1 \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Integration by parts

Basis: $\frac{(a+b \text{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)}$

Rule: If $e == c^2 d \land n < -1 \land (p \in \mathbb{Z} \lor d > 0)$, then

$$\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{d^p (1+c^2 x^2)^{p+\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)} - \frac{c d^p (2 p+1)}{b (n+1)} \int x (1+c^2 x^2)^{p-\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n+1} \, dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  d^p*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  c*d^p*(2*p+1)/(b*(n+1))*Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && LtQ[n,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

2: $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b \text{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)}$

Basis: If $e == c^2 d$, then $\partial_x \left(\sqrt{1+c^2 x^2}\, (d+e x^2)^p\right) == \frac{c^2 (2 p+1) x (d+e x^2)^p}{\sqrt{1+c^2 x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{(d+e x^2)^p}{(1+c^2 x^2)^p} == 0$

Rule: If $e == c^2 d \land n < -1$, then

$$\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx \to$$

$$\frac{\sqrt{1+c^2 x^2} (d+e x^2)^p (a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)} + \frac{c (2 p+1)}{b (n+1)} \int \frac{x (d+e x^2)^p (a+b \text{ArcSinh}[c x])^{n+1}}{\sqrt{1+c^2 x^2}} dx \to$$

$$\frac{\sqrt{1+c^2 x^2} (d+e x^2)^p (a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)} - \frac{c (2 p+1) (d+e x^2)^p}{b (n+1) (1+c^2 x^2)^p} \int x (1+c^2 x^2)^{p-\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n+1} dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  Simp[Sqrt[1+c^2*x^2]*(d+e*x^2)^p]*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)/(b*(n+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && LtQ[n,-1]
```

4: $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx$ when $e == c^2 d \land 2 p \in \mathbb{Z}^+$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $e == c^2 d$, then $\partial_x \frac{(d+e x^2)^p}{(1+c^2 x^2)^p} == 0$

Basis: $(1+c^2 x^2)^p == \frac{1}{b c} \text{Subst}\left[\text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2 p+1}, x, a+b \text{ArcSinh}[c x]\right] \partial_x (a+b \text{ArcSinh}[c x])$

Note: If $2 p \in \mathbb{Z}^+$, then $x^n \cosh\left[-\frac{a}{b}+\frac{x}{b}\right]^{2 p+1}$ is integrable in closed-form.

Rule: If $e == c^2 d \land 2 p \in \mathbb{Z}^+$, then

$$\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx \to \frac{(d+e x^2)^p}{(1+c^2 x^2)^p} \int (1+c^2 x^2)^p (a+b \text{ArcSinh}[c x])^n dx$$

$$\rightarrow \frac{\left(d+e x^2\right)^p}{b c \left(1+c^2 x^2\right)^p} \operatorname{Subst}\left[\int x^n \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2p+1} dx,\ x,\ a+b \operatorname{ArcSinh}[c x]\right]$$

– Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c)*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Subst[Int[x^n*Cosh[-a/b+x/b]^(2*p+1),x],x,a+b*ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && IGtQ[2*p,0]
```

2. $\int \left(d+e x^2\right)^p \left(a+b \operatorname{ArcSinh}[c x]\right)^n dx$ when $e \neq c^2 d$

1: $\int \left(d+e x^2\right)^p \left(a+b \operatorname{ArcSinh}[c x]\right) dx$ when $e \neq c^2 d \wedge \left(p \in \mathbb{Z}^+ \vee p+\frac{1}{2} \in \mathbb{Z}^-\right)$

– Derivation: Integration by parts

– Note: If $p \in \mathbb{Z}^+ \vee p+\frac{1}{2} \in \mathbb{Z}^-$, then $\int \left(d+e x^2\right)^p dx$ is a rational function.

– Rule: If $e \neq c^2 d \wedge \left(p \in \mathbb{Z}^+ \vee p+\frac{1}{2} \in \mathbb{Z}^-\right)$, let $u=\int \left(d+e x^2\right)^p dx$, then

$$\int \left(d+e x^2\right)^p \left(a+b \operatorname{ArcSinh}[c x]\right) dx \rightarrow u \left(a+b \operatorname{ArcSinh}[c x]\right) - b c \int \frac{u}{\sqrt{1+c^2 x^2}} dx$$

– Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[e,c^2*d] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

2: $\int (d + e x^2)^p (a + b \,\text{ArcSinh}[c x])^n \, dx$ when $e \neq c^2 d \wedge p \in \mathbb{Z} \wedge (p > 0 \vee n \in \mathbb{Z}^+)$

Derivation: Algebraic expansion

Rule: If $e \neq c^2 d \wedge p \in \mathbb{Z} \wedge (p > 0 \vee n \in \mathbb{Z}^+)$, then

$$\int (d + e x^2)^p (a + b \,\text{ArcSinh}[c x])^n \, dx \longrightarrow \int (a + b \,\text{ArcSinh}[c x])^n \,\text{ExpandIntegrand}\left[(d + e x^2)^p, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n,(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && NeQ[e,c^2*d] && IntegerQ[p] && (p>0 || IGtQ[n,0])
```

U: $\int (d + e x^2)^p (a + b \,\text{ArcSinh}[c x])^n \, dx$

Rule:

$$\int (d + e x^2)^p (a + b \,\text{ArcSinh}[c x])^n \, dx \longrightarrow \int (d + e x^2)^p (a + b \,\text{ArcSinh}[c x])^n \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x]
```

Rules for integrands of the form $(d+e x)^p (f+g x)^q (a+b \text{ArcSinh}[c x])^n$

1: $\int (d+e x)^p (f+g x)^q (a+b \text{ArcSinh}[c x])^n \, dx$ when $e f+d g==0 \land c^2 d^2+e^2==0 \land (p \mid q) \in \mathbb{Z}+\frac{1}{2} \land p-q \geq 0 \land d>0 \land \frac{g}{e}<0$

Derivation: Algebraic expansion

Basis: If $e f+d g==0 \land c^2 d^2+e^2==0 \land d>0 \land \frac{g}{e}<0$, then
$(d+e x)^p (f+g x)^q == \left(-\frac{d^2 g}{e}\right)^q (d+e x)^{p-q} \left(1+c^2 x^2\right)^q$

Rule: If $e f+d g==0 \land c^2 d^2+e^2==0 \land (p \mid q) \in \mathbb{Z}+\frac{1}{2} \land p-q \geq 0 \land d>0 \land \frac{g}{e}<0$, then

$$\int (d+e x)^p (f+g x)^q (a+b \text{ArcSinh}[c x])^n \, dx \to \left(-\frac{d^2 g}{e}\right)^q \int (d+e x)^{p-q} \left(1+c^2 x^2\right)^q (a+b \text{ArcSinh}[c x])^n \, dx$$

Program code:

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^q*Int[(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0] && GtQ[d,0] && LtQ[g/e,0]
```

2: $\int (d+e x)^p (f+g x)^q (a+b \text{ArcSinh}[c x])^n \, dx$ when $e f+d g==0 \land c^2 d^2+e^2==0 \land (p \mid q) \in \mathbb{Z}+\frac{1}{2} \land p-q \geq 0 \land \neg\left(d>0 \land \frac{g}{e}<0\right)$

Derivation: Piecewise constant extraction

Basis: If $e f+d g==0 \land c^2 d^2+e^2==0$, then $\partial_x \frac{(d+e x)^q (f+g x)^q}{\left(1+c^2 x^2\right)^q}==0$

Rule: If $e f+d g==0 \land c^2 d^2+e^2==0 \land (p \mid q) \in \mathbb{Z}+\frac{1}{2} \land p-q \geq 0 \land \neg\left(d>0 \land \frac{g}{e}<0\right)$, then

$$\int (d+e\,x)^p \left(f+g\,x\right)^q \left(a+b\,\text{ArcSinh}[c\,x]\right)^n \,dx \longrightarrow \frac{(d+e\,x)^q \left(f+g\,x\right)^q}{\left(1+c^2\,x^2\right)^q} \int (d+e\,x)^{p-q} \left(1+c^2\,x^2\right)^q \left(a+b\,\text{ArcSinh}[c\,x]\right)^n \,dx$$

Program code:

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^q*(f+g*x)^q/(1+c^2*x^2)^q*Int[(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0]
```